

Haskell

High-Order Functions

<http://igm.univ-mlv.fr/~vialette/?section=teaching>

Stéphane Vialette

LIGM, Université Paris-Est Marne-la-Vallée

November 20, 2017



Curried functions

Every function in Haskell officially takes only one parameter.

A curried function is a function that, instead of taking several parameters, always takes exactly one parameter.

When it is called with that parameter, it returns a function that takes the next parameter, and so on.



Curried functions

```
ghci> 1 + 2
3
ghci> :t +
<interactive>:1:1: parse error on input '+'
ghci> (+) 1 2
3
ghci> :t (+)
(+) :: Num a => a -> a -> a
ghci>
```




Curried functions

```
ghci> let add = (+)
ghci> :t add
add :: Num a => a -> a -> a
ghci> add 1 2
3
ghci> (add 1) 2
3
ghci> let add1 = add 1
ghci> :t add 1
add 1 :: Num a => a -> a
ghci> add1 2
3
ghci>
```



Curried functions

Whenever we have a type signature that features the arrow \rightarrow , that means it is a function that takes whatever is on the left side of the arrow and returns a value whose type is indicated on the right side of the arrow.



Curried functions

When we have something like $a \rightarrow a \rightarrow a$, we are dealing with a function that takes a value of type a , and it returns a function that also takes a value of type a and returns a value of type a .

In other words $a \rightarrow a \rightarrow a$ reads as $a \rightarrow (a \rightarrow a)$.



Curried functions

```
ghci> let multThree x y z = x * y * z
ghci> :t multThree
multThree :: Num a => a -> a -> a -> a
ghci> let multTwoWithNine = multThree 9
ghci> :t multTwoWithNine
multTwoWithNine :: Num a => a -> a -> a
ghci> multTwoWithNine 2 3
54
ghci> let multWithNineAndFive = multTwoWithNine 5
ghci> :t multWithNineAndFive
multWithNineAndFive :: Num a => a -> a
ghci> multWithNineAndFive 2
90
ghci> multThree 2 5 9
90
ghci>
```



Curried functions

```
ghci> :t compare
compare :: Ord a => a -> a -> Ordering
ghci> :t (compare 100)
(compare 100) :: (Ord a, Num a) => a -> Ordering
ghci> let compareWithHundred x = compare 100 x
ghci> compareWithHundred 99
GT
ghci> :t compareWithHundred
compareWithHundred :: (Ord a, Num a) => a -> Ordering
ghci> let compareWithHundred' = compare 100
ghci> :t compareWithHundred'
compareWithHundred' :: (Ord a, Num a) => a -> Ordering
ghci> compareWithHundred' 99
GT
ghci>
```



Curried functions

```
ghci> let divideByTen = (/10)
ghci> :t divideByTen
divideByTen :: Fractional a => a -> a
ghci> divideByTen 200
20.0
ghci> (/ 10) 200
20.0
ghci> let isUpperAlphanum = (`elem` ['A'..'Z'])
ghci> :t isUpperAlphanum
isUpperAlphanum :: Char -> Bool
ghci> isUpperAlphanum 'k'
False
ghci> isUpperAlphanum 'K'
True
ghci>
```



Some Higher-Orderism Is in Order

In Haskell, function can take other functions as parameter, and as we have seen, they can also return functions as return value.

```
applyTwice :: (a -> a) -> a -> a
applyTwice f x = f (f x)
```

`->` is naturally right-associative. Therefore, here parentheses are mandatory as `a -> a -> a -> a` is interpreted by Haskell as `a -> (a -> (a -> a))`.



Some Higher-Orderism Is in Order

```
ghci> applyTwice (+3) 10
16
ghci> (+3) ((+3) 10)
16
ghci> applyTwice (++ " HAHA") "HEY"
"HEY HAHA HAHA"
ghci> applyTwice ("HAHA " ++ ) "HEY"
"HAHA HAHA HEY"
ghci> let multThree x y z = x * y * z
      in applyTwice (multThree 2 2) 9
144
ghci> applyTwice (1:) [2]
[1,1,2]
ghci>
```



First-class and higher-order functions

I ONLY FLY



FIRST-CLASS FUNCTIONS



Some Higher-Orderism Is in Order

Implementing `zipWith`

`zipWith` takes a function and two lists as parameters, and then joins the two lists by applying the function between corresponding elements (it's in the standard library).

```
zipWith' :: (a -> b -> c) -> [a] -> [b] -> [c]
zipWith' _ [] _ = []
zipWith' _ _ [] = []
zipWith' f (x:xs) (y:ys) = f x y : zipWith' f xs ys
```



Some Higher-Orderism Is in Order

Implementing `zipWith`

```
ghci> :t zipWith'
zipWith' :: (a -> b -> c) -> [a] -> [b] -> [c]
ghci> zipWith' (+) [1,2,3] [11,12,13]
[12,14,16]
ghci> zipWith' max [1,12,3] [11,2,13]
[11,12,13]
ghci> zipWith' (++) ["foo","bar"] ["fighter","hoppers"]
["foofighter","barhoppers"]
ghci> zipWith' (*) (replicate 5 2) [1..]
[2,4,6,8,10]
ghci> zipWith' (zipWith' (*)) [[1,2],[3,4]] [[5,6],[7,8]]
[[5,12],[21,32]]
ghci>
```



Some Higher-Orderism Is in Order

Implementing `flip`

`flip` takes a function and returns a function that is like our original function, but with the first two arguments flipped (it's in the standard library).

```
flip' :: (a -> b -> c) -> b -> a -> c
flip' f = g where g x y = f y x
```

Recall that the arrow `->` is right-associative, and hence

`(a -> b -> c) -> b -> a -> c` is the same as
`(a -> b -> c) -> (b -> a -> c)`.

```
flip' :: (a -> b -> c) -> b -> a -> c
flip' f x y = f y x
```




Some Higher-Orderism Is in Order

Implementing `flip`

```
ghci> zip [1..5] "hello"  
[(1, 'h'), (2, 'e'), (3, 'l'), (4, 'l'), (5, 'o')]  
ghci> flip' zip [1..5] "hello"  
[('h', 1), ('e', 2), ('l', 3), ('l', 4), ('o', 5)]  
ghci> zipWith div [2,2..] [10,8,6,4,2]  
[0,0,0,0,1]  
ghci> zipWith (flip' div) [2,2..] [10,8,6,4,2]  
[5,4,3,2,1]  
ghci>
```



The functional Programmer's Toolbox



The functional Programmer's Toolbox

The `map` function

The `map` function takes a function and a list, and applies that function to every element in the list, producing a new list.

```
map :: (a -> b) -> [a] -> [b]
```

```
map _ [] = []
```

```
map f (x:xs) = f x : map f xs
```

`map` is a versatile higher-order function that can be used in many different ways



The functional Programmer's Toolbox

The `map` function

```
ghci> map (+1) [1,2,3,4,5]
[2,3,4,5,6]
ghci> map (++ "!") ["BIFF", "BANG", "POW"]
["BIFF!", "BANG!", "POW!"]
ghci> map (replicate 3) [1,2,3]
[[1,1,1], [2,2,2], [3,3,3]]
ghci> map (map (^2)) [[1,2], [3,4]]
[[1,4], [9,16]]
ghci> map fst [(1,2), (3,4), (5,6)]
[1,3,5]
ghci> map snd [(1,2), (3,4), (5,6)]
[2,4,6]
ghci>
```



The functional Programmer's Toolbox

The `filter` function

The `filter` function takes a predicate and a list, and returns the list of elements that satisfy the predicate

```
filter :: (a -> Bool) -> [a] -> [a]
filter _ []      = []
filter p (x:xs)
  | p x          = x : filter p xs
  | otherwise    = filter p xs
```

If `p x` evaluates to `True`, the element is included in the new list. If it doesn't evaluate to `True`, it isn't included in the new list.



The functional Programmer's Toolbox

The `filter` function

```
ghci> filter (> 3) [1,2,3,4,5,1,2,3,4,5]
[4,5,4,5]
ghci> filter (== 3) [1,2,3,4,5,1,2,3,4,5]
[3,3]
ghci> filter (< 3) [1,2,3,4,5,1,2,3,4,5]
[1,2,1,2]
ghci> filter even [1,2,3,4,5,1,2,3,4,5]
[2,4,2,4]
ghci> filter (`elem` ['a'..'z']) "I l0vE hAsKeLl"
"lvhsel"
ghci> filter (`elem` ['A'..'Z']) "I l0vE hAsKeLl"
"IOEAKL"
ghci>
```



The functional Programmer's Toolbox

The `filter` function

The `filter` equivalent of applying several predicates in a list comprehension is either filtering something several times or joining predicates with the logical `&&` function.

```
ghci> filter (< 15) (filter even [1..20])
[2,4,6,8,10,12,14]
ghci> let p x = x < 15 && even x in filter p [1..20]
[2,4,6,8,10,12,14]
ghci> filter (\x -> x < 15 && even x) [1..20]
[2,4,6,8,10,12,14]
ghci> [x | x <- [1..20], x < 15, even x]
[2,4,6,8,10,12,14]
ghci>
```



The functional Programmer's Toolbox

The `filter` function

```
quicksort :: (Ord a) => [a] -> [a]
quicksort []      = []
quicksort (x:xs) =
    let smallerOrEqual = filter (<= x) xs
        larger         = filter (> x)  xs
    in quicksort smallerOrEqual ++ [x] ++ quicksort larger
```



The functional Programmer's Toolbox

More examples of `map` and `filter`

Let's find the largest number under 100 000 that is divisible by 3829.

Think-About



The functional Programmer's Toolbox

More examples of `map` and `filter`

Let's find the largest number under 100 000 that is divisible by 3829.

```
largestDivisible :: Integer
largestDivisible = head (filter p [100000,99999..])
  where p x = x `mod` 3829 == 0
```



The functional Programmer's Toolbox

More examples of `map` and `filter`

Let's find the sum of all odd squares that are smaller than 10 000.

Think-About



The functional Programmer's Toolbox

More examples of `map` and `filter`

Let's find the sum of all odd squares that are smaller than 10 000.

```
ghci> sum (takeWhile (< 10000) (filter odd (map (^2) [1..])))
166650
ghci> sum (takeWhile (< 10000) [x | x <- [y^2 | y <- [1..]],
                                odd x])
166650
ghci>
```



The functional Programmer's Toolbox

More examples of `map` and `filter`

A *Collatz* sequence is defined as follows:

- Start with any natural number.
- If the number is 1, stop.
- If the number is even, divide it by 2.
- If the number is odd, multiply it by 3 and add 1.
- Repeat the algorithm with the resulting number.

Mathematicians theorize that for all starting number, the chain will finish at the number 1.



The functional Programmer's Toolbox

More examples of `map` and `filter`


$$F(n) = \begin{cases} \frac{n}{2} & \text{if } n \% 2 = 0 \\ 3n + 1 & \text{if } n \% 2 = 1 \end{cases}$$



The functional Programmer's Toolbox

More examples of `map` and `filter`

Two Collatz Sequences



The functional Programmer's Toolbox

More examples of `map` and `filter`

```
collatz :: Integer -> [Integer]
collatz 1 = [1]
collatz n
  | even n = n : collatz (n `div` 2)
  | odd n  = n : collatz (n*3 + 1)
```

```
ghci> collatz 10
[10,5,16,8,4,2,1]
ghci> collatz 20
[20,10,5,16,8,4,2,1]
ghci> length $ collatz 100
26
ghci> length $ collatz 1000
112
ghci>
```



The functional Programmer's Toolbox

Mapping functions with Multiple Parameters

```
ghci> let listOfFuns = map (*) [0..]
ghci> :t listOfFuns
listOfFuns :: (Num a, Enum a) => [a -> a]
ghci> (listOfFuns !! 0) 5
0
ghci> (listOfFuns !! 1) 5
5
ghci> (listOfFuns !! 2) 5
10
ghci> (listOfFuns !! 3) 5
15
ghci> (listOfFuns !! 4) 5
20
ghci>
```



Lambdas



Lambdas

Lambdas are anonymous functions that we use when we need a function only once.

Normally, we make a lambda with the sole purpose of passing it to a higher-order function.

To declare a lambda, we write `\` (because it kind of looks like the Greek letter lambda (λ) if you squint hard enough), and then we write the function's parameters, separated by spaces.

After that comes a `->`, and then the function body.

If a lambda match fails in a lambda, a runtime error occurs, so be careful!



Lambdas

```
ghci> map (+3) [1..5]
[4,5,6,7,8]
ghci> map (\x -> x + 3) [1..5]
[4,5,6,7,8]
ghci> zipWith (+) [1..5] [101..105]
[102,104,106,108,110]
ghci> zipWith (\x y -> x + y) [1..5] [101..105]
[102,104,106,108,110]
ghci> map (\(x,y) -> x + y) [(1,2),(3,4),(5,6)]
[3,7,11]
ghci>
```



Lambdas



Lambdas

The following functions are equivalent:

```
addThree :: Int -> Int -> Int -> Int
addThree x y z = x + y + z
```

```
addThree' :: Int -> Int -> Int -> Int
addThree' = \x -> \y -> \z -> x + y + z
```

In the second example, the lambdas are not surrounded with parentheses. When you write a lambda without parentheses, it assumes that everything to the right of the arrow \rightarrow belongs to it.



Lambdas

The following functions are equivalent:

```
flip' :: (a -> b -> c) -> b -> a -> c
```

```
flip' f x y = f y x
```


```
flip'' :: (a -> b -> c) -> b -> a -> c
```

```
flip'' f = \x y -> f y x
```

In the second example, our new notation makes it obvious that this will often be used for producing a new function.



I fold you so




I fold you so

- Folds can be used to implement any function where you traverse a list once, element by element, and then return something based on that.
- A fold takes a *binary function* (one that takes two parameters, such as + or `div`), a starting value (often called the *accumulator*), and a list to fold up.
- Lists can be folded up from the left or from the right.
- The fold function calls the given binary function, using the accumulator and the first (or last) element of the list as parameters. The resulting value is the new accumulator.
- The accumulator value (and hence the result) of a fold can be of any type.



Left fold with `foldl`



Lambdas



I fold you so

Left fold with `foldl`

```
sum' :: (Num a) => [a] -> a
```

```
sum' xs = foldl (\acc x -> acc + x) 0 xs
```

```
ghci> sum' []
```

```
0
```

```
ghci> sum' [3,5,2,1]
```

```
11
```

```
ghci>
```



I fold you so

Left fold with `foldl`

$$0 + 3$$

$$[3, 5, 2, 1]$$

$$3 + 5$$

$$[5, 2, 1]$$

$$8 + 2$$

$$[2, 1]$$

$$10 + 1$$

$$[1]$$

11



I fold you so

Left fold with `foldl`

The lambda function `\acc x -> acc + x` is the same as `(+)`

```
sum'' :: (Num a) => [a] -> a
```

```
sum'' = foldl (+) 0
```

```
ghci> sum'' []
```

```
0
```

```
ghci> sum'' [3,5,2,1]
```

```
11
```

```
ghci>
```



A quick parenthesis

η -reduction

- An **eta conversion** (also written η -**conversion**) is adding or dropping of abstraction over a function.
- For example, the following two values are equivalent under η -conversion: $\lambda x \rightarrow \text{abs } x$ and **abs**.
- Converting from the first to the second would constitute an η -**reduction**, and moving from the second to the first would be an η -**abstraction**.
- The term η -**conversion** can refer to the process in either direction.
- Extensive use of η -reduction can lead to **Pointfree programming**.



A quick parenthesis

η -reduction

Therefore

```
sum'' :: (Num a) => [a] -> a
sum'' xs = foldl (+) 0 xs
```

is usually rewritten as:

```
sum'' :: (Num a) => [a] -> a
sum'' = foldl (+) 0
```



I fold you so

Left fold with `foldl`

```
elem' :: (Eq a) => a -> [a] -> Bool
```

```
elem' y ys = foldl (\acc x -> if x == y then True else acc)  
                  False ys
```

```
Prelude> elem' 'a' ['a'..'l']
```

```
True
```

```
Prelude> elem' 'm' ['a'..'l']
```

```
False
```

```
Prelude> elem' (3, 9) [(i, i^2) | i <- [1..100]]
```

```
True
```

```
Prelude> elem' (4, 17) [(i, i^2) | i <- [1..100]]
```

```
False
```

```
Prelude>
```





I fold you so

Right fold with `foldr`

- The right fold function `foldr` is similar to the left fold, except that the accumulator eats up the values from the right.
- Also, the order of the parameters in the right fold's binary function is reversed: The current list value is the right parameter and the accumulator is the second.



Right fold with `foldr`



I fold you so

Right fold with `foldr`

```
map' :: (a -> b) -> [a] -> [b]
```

```
map' f xs = foldr (\x acc -> f x : acc) [] xs
```

```
ghci> map' (+ 10) []
```

```
[]
```

```
ghci> map' (+ 10) [1..5]
```

```
[11,12,13,14,15]
```

```
ghci>
```



I fold you so

Right fold with `foldr`

```
map'  :: (a -> b) -> [a] -> [b]
map'  f = foldr (\x acc -> f x : acc) []
```

```
map'' :: (a -> b) -> [a] -> [b]
map'' f = foldl (\acc x -> acc ++ [f x]) []
```

Notice that the `++` function is much slower than `:`, so we usually use right fold when we are building up new lists from lists.



I fold you so

Right fold with `foldr`

The `elem` function checks whether a value is part of a list.

```
elem' :: (Eq a) => a -> [a] -> Bool
elem' x = foldr (\y acc -> if x==y then True else acc)
              False
```

```
ghci> :t elem'
elem' :: Eq a => a -> [a] -> Bool
ghci> 5 `elem` [10..20]
False
ghci> 15 `elem` [10..20]
True
ghci>
```



I fold you so

The `foldl1` and `foldr1` functions

- The `foldl1` and `foldr1` functions work much like `foldl` and `foldr`, except that you don't need to provide them with an explicit starting accumulator.
- The `foldl1` and `foldr1` functions assume the first (or last) element of the list to be the starting accumulator, and then start the fold with the next element next to it.



I fold you so

The `foldl1` and `foldr1` functions

```
ghci> :t foldl1
foldl1 :: (a -> a -> a) -> [a] -> a
ghci> :t foldr1
foldr1 :: (a -> a -> a) -> [a] -> a
ghci>
```



I fold you so

The `foldl1` and `foldr1` functions

```
minimum' :: (Ord a) => [a] -> a  
minimum' = foldl1 min
```

```
maximum' :: (Ord a) => [a] -> a  
maximum' = foldl1 max
```

```
ghci> :t minimum'  
minimum' :: Ord a => [a] -> a  
ghci> minimum' []  
*** Exception: Prelude.foldl1: empty list  
ghci> minimum' [1]  
1  
ghci> minimum' $ [10..20] ++ [1..10]  
1  
ghci>
```



I fold you so

Some fold examples

```
reverse' :: [a] -> [a]
reverse' = foldl (\acc x -> x : acc) []
```

```
reverse'' :: [a] -> [a]
reverse'' = foldl (flip (:)) []
```

```
Main> reverse' []
[]
ghci> reverse' [1..5]
[5,4,3,2,1]
ghci> reverse'' []
[]
ghci> reverse'' [1..5]
[5,4,3,2,1]
ghci>
```



I fold you so

Some fold examples

```
filter' :: (a -> Bool) -> [a] -> [a]
```

```
filter' p = foldr (\x acc -> if p x then x : acc else acc)  
                []
```

```
last' :: [a] -> a
```

```
last' = foldl1 (\_ x -> x)
```

```
length' :: Num b => [a] -> b
```

```
length' = foldr (\_ -> (+1)) 0
```



I fold you so

Folding infinite lists

```
and' :: [Bool] -> Bool  
and' = foldr (&&) True
```

```
ghci> and' (repeat False)  
False  
ghci>
```



I fold you so

Scans

- The `scanl` and `scanr` functions are like `foldl` and `foldr`, except they report all the intermediate accumulator states in the form of a list.
- The `scanl1` and `scanr1` functions are analogous to `foldl1` and `foldr1`.



I fold you so

Scans

```
ghci> scanl (+) 0 [1,2,3,4]
[0,1,3,6,10]
ghci> scanr (+) 0 [1,2,3,4]
[10,9,7,4,0]
ghci> scanl1 (\acc x -> if x > acc then x else acc) [1..5]
[1,2,3,4,5]
ghci> scanl1 max [1..5]
[1,2,3,4,5]
ghci> scanl (flip (:)) [] [3,2,1]
[[], [3], [2,3], [1,2,3]]
ghci>
```



I fold you so

Function application with \$

The *function application operator* \$ is defined as follows:

$(\$)$:: $(a \rightarrow b) \rightarrow a \rightarrow b$

$f \$ x = f x$



I fold you so

Function application with \$



I fold you so

Function application with $\$$

What is this useless function? It is just function application! Well, that is almost true, but not quite.

Whereas normal function application (putting a space between two things) has a really high precedence, the $\$$ function has the lowest precedence.

Function application with a space is left-associative (so $f\ a\ b\ c$ is the same as $((f\ a)\ b)\ c$), while function application with $\$$ is right-associative.



I fold you so

Function application with \$

```
ghci> sum (filter (> 10) (map (*2) [2..10]))
80
ghci> sum $ filter (> 10) (map (*2) [2..10])
80
ghci> sum $ filter (> 10) $ map (*2) [2..10]
80
ghci>
```



I fold you so

Function application with \$

Apart of getting rid of parentheses, \$ let us treat function application like just another function.

```
ghci> :t (4+)
(4+) :: Num a => a -> a
ghci> :t (^2)
(^2) :: Num a => a -> a
ghci> :t sqrt
sqrt :: Floating a => a -> a
ghci> :t [(4+),(^2),sqrt]
[(4+),(^2),sqrt] :: Floating a => [a -> a]
ghci> map ($) 3 [(4+),(^2),sqrt]
[7.0,9.0,1.7320508075688772]
ghci>
```




Function composition

In mathematics, *function composition* is defined as follows:

$$(f \circ g)(x) = f(g(x))$$

Composed Functions



Function composition

In Haskell, function composition is pretty much the same thing.

We do function composition with the `.` function:

```
(.) :: (b -> c) -> (a -> b) -> (a -> c)  
f . g = \x -> f (g x)
```



Function composition

```
ghci> :t negate
negate :: Num a => a -> a
ghci> :t abs
abs :: Num a => a -> a
ghci> map (\x -> negate(abs x)) [1,-2,3,-4,5,-6]
[-1,-2,-3,-4,-5,-6]
ghci> map (negate . abs) [1,-2,3,-4,5,-6]
[-1,-2,-3,-4,-5,-6]
ghci>
```



Function composition

```
ghci> map (\xs -> negate (sum (tail xs))) [[1..5],[3..6]]
[-14,-15]
ghci> map (negate . sum . tail) [[1..5],[3..6]]
[-14,-15]
ghci>
```

`negate . sum . tail` is a function that takes a list, applies the `tail` function to it, then applies the `sum` function to the result, and finally applies `negate` to the previous result.



Function composition

Function Composition with Multiple Parameters

But what about functions that take several parameters?

Well, if we want to use them in function composition, we usually must partially apply them so that each function takes just one parameter.

```
ghci> sum (replicate 5 (max 6.7 8.9))
44.5
ghci> (sum .replicate 5) (max 6.7 8.9)
44.5
ghci> sum .replicate 5 $ max 6.7 8.9
44.5
ghci> replicate 2 (product (map (*3) (zipWith max [1,2] [4,5])))
[180,180]
ghci> replicate 2 . product . map (*3) $ zipWith max [1,2] [4,5]
[180,180]
ghci>
```



Function composition

Point-Free Style

Another common use of function composition is defining function in the *point-free style*.

```
f :: (RealFrac a, Integral b, Floating a) => a -> b
f x = ceiling (negate (tan (cos (max 50 x))))
```

```
f' :: (RealFrac a, Integral b, Floating a) => a -> b
f' = ceiling . negate . tan . cos . max 50
```



Function composition

Point-Free Style

```
oddSquareSum :: Integer
oddSquareSum = sum (takeWhile (<10000) (filter odd (map (^2) [1..])))
```

```
oddSquareSum' :: Integer
oddSquareSum' = sum . takeWhile (<10000) . filter odd . map (^2) $ [1..]
```

```
oddSquareSum'' :: Integer
oddSquareSum'' =
  let oddSquares = filter odd $ map (^2) [1..]
      belowLimit = takeWhile (<10000) oddSquares
  in sum belowLimit
```

